
1. 目的 
 
・ UNIX ワークステーションの環境を理解し利用できる． 
・ メールの送受信，telnet，ftp を使える． 
・ 基本コマンド，ls，mv，rm，cp，cd，grep，cat，less，．．．を使える． 
・ エディタ，vi，emacs，．．．を使える． 
・ FORTRAN でプログラムを作成できる． 
・ GNUPLOT で図を作成できる． 
・ なぜ数値計算が必要か理解する． 
 
2. 内容 
 
連立１次方程式と逆行列 
 消去法（ガウス法） 
 反復法（ガウス・ザイデル法） 
 逆行列 
代数方程式 
 ニュートン法 
 はさみうち法 
数値積分 
 台形公式 
 シンプソンの公式 
フーリエ級数 
常微分方程式 
 ルンゲ・クッタ法 
 連立微分方程式 
ポアソン方程式・ラプラス方程式 
 差分法 
偏微分方程式 
 熱伝導の方程式 
 波動方程式 
 
3．On WEB 
 

http://www.ep.sci.hokudai.ac.jp/~shw/computer.pdf 

UNIX基本コマンド 

コマンド  機能     使用例 

ls     ls, ls –l, ls –al 

cp コピー    cp file1 file2 

mv ファイル名の変更   mv file1 file2 

rm ファイルやディレクトリの削除 rm file1, rm –r direct1 

less     less file1 

tar アーカイブの解凍・作成  tar tvf file1.tar, tar xvf file1.tar 

mkdir ディレクトリの作成  mkdir direct2 

cd ディレクトリの変更  cd, cd /usr/local 

pwd 現ディレクトリの表示  pwd 

find ファイルやディレクトリの検索 find / -name file1 

grep 文字検索   grep name file1 

telnet     telnet comp1.hokudai.ac.jp 

ftp     ftp comp1.hokudai.ac.jp 

     ftp –i comp1.hokudai.ac.jp 

 

 

 

emacs の基本操作 

起動方法     emacs filename 

新しいウィンドウを作らずに起動  emacs –nw filename 

カーソルの移動    上Ctrl-+p 下Ctrl+n 左Ctrl+b 右Ctrl+f 

1 画面後ろへ    Ctrl+v 

1 画面前へ    Esc+v 

1 文字削除    Ctrl+d 

カーソルから行の最後まで削除  Ctrl+k 

移動とコピー    Ctrl+y 

ファイルの後方へサーチ   Ctrl+s string 

ファイルへ保存    Ctrl+x Ctrl+s 

ファイル名を指定して保存   Ctrl+x Ctrl+w filename 

コマンドの実行中止   Ctrl+g 

終了     Ctrl+x Ctrl+c 

vi の基本操作 

起動方法     vi filename 

カーソルの移動    上 j 下 k 左 h 右 l 

1 画面後ろへ    Ctrl+f 

1 画面前へ    Ctrl+b 

1 文字削除    x 

行の削除     dd 

コピー     yy 

ペースト     p 

挿入     i 

アペンド     a 

ファイルの後方へサーチ   /string 

ファイルへ保存    :w 

ファイル名を指定して保存   :w filename 

コマンドの実行モードへ   ESC 

終了     :q 

セーブして終了    :wq 

変更しないで終了    :q! 

 

 

 

FORTRANプログラムのコンパイル  f77 file1.f 

 f77 file1.f –o file1 

g77 file1.f 

 g77 file1.f –o file1 

  g90 file1.f 

 g90 file1.f –o file1 

 

実行      ./a.out 

 ./file1 

 ./a.out > outfile 

      ./file1 > outfile 

      ./a.out < infile 

 

GNUPLOTプログラムの実行   gnuplot file2 

課題１ 連立１次方程式 
 

1332 4321 =−++ xxxx  

6422 4321 =+++− xxxx  

3534 4321 =+−+ xxxx  

3445 4321 =+−− xxxx  

 
をガウスの消去法で解け． 
 
 
 
（解法）n元連立１次方程式は次のように表すことができる． 

１列を 11a で割ると 

１列を 21a 倍して２列から引くと 

同様にして計算し，最終的に 

となったとき， nn AAAA ,,..., 121 − が答えである． 

 
 
 
（FORTRANプログラム例） 
 
c    Gauss method 
c 
      real*4 a(4,5) 
c 
      a(1,1) =  2.0 
      a(1,2) =  3.0 
      a(1,3) =  1.0 
      a(1,4) = -3.0 
      a(1,5) =  1.0 
      a(2,1) = -1.0 
      a(2,2) =  2.0 
      a(2,3) =  2.0 
      a(2,4) =  4.0 
      a(2,5) =  6.0 
      a(3,1) =  4.0 
      a(3,2) =  1.0 
      a(3,3) = -3.0 
      a(3,4) =  5.0 
      a(3,5) =  3.0 
      a(4,1) =  5.0 
      a(4,2) = -4.0 
      a(4,3) = -4.0 
      a(4,4) =  1.0 
      a(4,5) =  3.0 
c 
      do k=1,4 
         p = a(k,k) 
         do j=k,5 

            a(k,j) = a(k,j)/p 
         enddo 
         do i=1,4 
            if( i .ne. k ) then 
                aik = a(i,k) 
                do j=k,5 
                   a(i,j) = a(i,j) - aik*a(k,j) 
                enddo 
            endif 
         enddo 
      enddo 
      do i=1,4 
         write(6,*) i,'   ',(a(i,j),j=1,5) 
      enddo 
      stop 
      end 
 
 
 
（実行結果） 
 
1     1.  0.  0.  0.  1.99999952 
2     0.  1.  0.  0. -0.999999523 
3     0.  0.  1.  0.  2.99999928 
4     0.  0.  0.  1.  0.999999881 

課題２ 連立１次方程式 
 

44 321 =+− xxx  

926 321 =++ xxx  

252 321 =+−− xxx  

 
をガウス・ザイデルの反復法で解け． 
 
（解法）n元連立１次方程式は次のように変形することができる． 

で判定する．対角要素が他の要素より大きいとき，ガウス・ザイデル法は収束する． 
 
 
消去法 
・方程式の数で演算回数が決まる． 
・誤差が積み重ねられる． 
反復法 
・収束しない場合がある． 
・演算回数が多い． 
・０を多く含んだ方程式では演算回数が大幅に減り，有効である． 
 
 
（FORTRANプログラム例） 
c   EX2 
c   Gauss-Seidel method 
c 
      real*4  x(3), xb(3) 
      real*4  a(3,4) 
c 
      read(5,*) ((a(i,j),j=1,4),i=1,3) 

c 
      do i=1,3 
         x(i) = 0.0 
      enddo 
c 
      do n=1,100 
         do i=1,3 
            xb(i) = x(i) 
            x(i) = a(i,4)/a(i,i) 
            do j=1,3 
               if( i .ne. j ) then 
                   x(i) = x(i) - a(i,j)*x(j)/a(i,i) 
               endif 
            enddo 
         enddo 
         ic = 0 
         do i=1,3 
            if( abs(x(i)-xb(i)) .lt. 1.0e-6 ) ic=ic+1 
         enddo 
         write(6,*) n,x 
         if( ic .eq. 3 ) go to 200 
      enddo 
  200 continue 
      write(6,*) 'Answer' 
      write(6,*) x 
      stop 
      end 
 
（入力データ） 
4.0  -1.0  1.0  4.0 
1.0   6.0  2.0  9.0 
-1.0  -2.0  5.0  2.0 
 
（実行結果） 
1  1.  1.33333337  1.13333333 
2  1.05000007  0.947222292  0.988888919 
3  0.989583313  1.00543976  1.00009251 
4  1.00133681  0.999746382  1.00016594 
5  0.999895096  0.999962151  0.99996388 
6  0.999999642  1.00001204  1.00000477 
7  1.00000179  0.999998093  0.999999642 
8  0.999999642  1.00000012  0.99999994 
9  1.  1.  1. 
 Answer 
1.  1.  1. 

課題３ 

の逆行列を求めよ． 
 
 
 

（解法） 0≠A の行列 Aに対して， IAB = （ I は単位行列）を満たす行列 B を Aの逆行

列という． Aの逆行列を 1−A とすると 

となる．したがって， 

の３元連立１次方程式の３つの組から ijx を求めることになる．これは，課題１の方法と同

様のアルゴリズムで解くことができる．すなわち， 

と変形した時， 

が逆行列である． 

（FORTRANプログラム例） 
c 
c       e x -3 
c 
      real*4 a(10,20) 
c 
      read(5,*) n,((a(i,j),j=1,n),i=1,n) 
      do i=1,n 
         do j=n+1,2*n 
            i f (  i  .eq.  ( j -n) ) then 
                a ( i , j )  =  1 .0  
            e l s e  
                a ( i , j )  =  0 .0  
            e n d i f  
         e n d d o  
      enddo  
      write(6,*) ' Input Data ' 
      write(6,800) ((a(i,j),j=1,2*n),i=1,n) 
  800 format(6f5.1) 
c 
      do k=1,n 
         p = a(k,k)  
         do j=k,2*n 
            a(k,j)  = a(k,j) /p 
         e n d d o  
         do  i=1,n 
            if( i  .ne. k ) then 
                a ik  =  a ( i ,k )  
                do  j=k ,2*n  
                   a(i,j) = a(i,j) - aik*a(k,j) 
                e n d d o  
            e n d i f  
         e n d d o  
      enddo  
c 
      write(6,*) ' Inverse Matrix ' 
      write(6,810) ((a(i,j),j=n+1,2*n),i=1,n) 
  810 format(3e13.4) 
c 
      s top  
      e n d  

（入力データ） 
3 
2 3 4 
3 5 2 
4 3 30 
 
 
 
（実行方法） 
 
f90 ex3.f 
./a.out  <  ex3.dat 
 
 
 
（実行結果） 
 
  Input Data 
  2.0  3.0  4.0  1.0  0.0  0.0 
  3.0  5.0  2.0  0.0  1.0  0.0 
  4.0  3.0 30.0  0.0  0.0  1.0 
  Inverse Matrix 
  -0.7200E+02   0.3900E+02   0.7000E+01 
   0.4100E+02  -0.2200E+02  -0.4000E+01 

0.5500E+01  -0.3000E+01  -0.5000E+00 
 
 

(課題＋α) 

課題１のプログラムで n行m列の入力データをファイルから読み込

み計算するプログラムに変更せよ． 

課題４ 
 

0133)( 23 =+++= xxxxf  

 

の根を Newton 法で求めよ．ただし， 00 =x とし，２つの近似根間の相対差 510−=ε を収

束判定条件とする． 
 
 
 

（解法）関数 )(xf が単調連続で変曲点がなく，かつ )(xf の導関数が求められるとき，

0)( =xf の近似根を 0x ，真の根を hx +0 とすると， 0)( 0 =+ hxf である．テーラー

（Taylor）展開すると， 

となる． 

のとき nx を解とする 

 収束判定の条件として，以下のような種類がある． 
（１）２つの近似根間の差 

ε<−+ nn xx 1  

（２）２つの近似根間の相対差 

n

nn

x
xx 1  

（３）２つの近似根の関数値間の相対差 

（FORTRANプログラム例） 
 
c 
c      ex4 
c 
      x = 0.0 
      do i=1,1000 
         fx = x**3 + 3.0*x**2 + 3.0*x + 1.0 
         fxd = 3.0*x**2 + 6.0*x + 3.0 
         dx = -fx/fxd 
         write(6,*) i,x,fx,fxd,dx 
         x = x + dx 
         if( abs(dx/x) .lt. 1.0e-5 ) then 

             stop 
         endif 
      enddo 
      stop 
      end 
 
 
（実行結果） 
 
1  0.  1.  3. -0.333333343 
 2 -0.333333343  0.296296269  1.33333325 -0.222222209 
 3 -0.555555582  0.0877914801  0.592592537 -0.148148134 
 4 -0.703703701  0.0260122959  0.263374478 -0.0987654403 
 5 -0.802469134  0.0077073467  0.117055327 -0.0658436194 
 6 -0.868312776  0.0022836572  0.0520245731 -0.0438957401 
 7 -0.912208498  0.000676639669  0.0231220443 -0.0292638335 
 8 -0.941472352  0.000200485621  0.0102764573 -0.0195092149 
 9 -0.960981548  5.94032354E-05  0.0045673186 -0.0130061507 
 10 -0.973987699  1.76009598E-05  0.00202991953 -0.00867076684 
 11 -0.982658446  5.21511674E-06  0.000902188476 -0.00578051805 
 12 -0.988438964  1.54521979E-06  0.000400972669 -0.0038536787 
 13 -0.992292643  4.57842901E-07  0.000178210073 -0.00256911898 
 14 -0.994861782  1.35655583E-07  7.92038627E-05 -0.00171273947 
 15 -0.996574521  4.01942479E-08  3.52017196E-05 -0.00114182627 
 16 -0.997716367  1.19090959E-08  1.5644935E-05 -0.000761210977 
 17 -0.998477578  3.52862095E-09  6.95330482E-06 -0.000507473946 
 18 -0.998985052  1.04551734E-09  3.09035772E-06 -0.000338315964 
 19 -0.999323368  3.09782922E-10  1.3734923E-06 -0.000225543976 
 20 -0.999548912  9.1787529E-11  6.1044102E-07 -0.000150362655 
 21 -0.999699295  2.71909145E-11  2.71271261E-07 -0.000100235149 
 22 -0.99979955  8.05417191E-12  1.20541117E-07 -6.68167995E-05 
 23 -0.999866366  2.38642136E-12  5.3573828E-08 -4.45445366E-05 
24 -0.999910891  7.07560963E-13  2.38212117E-08 -2.97029801E-05 
 25 -0.999940574  2.09858173E-13  1.05942881E-08 -1.98086145E-05 
 26 -0.999960363  6.22737539E-14  4.71329642E-09 -1.32123569E-05 
 27 -0.999973595  1.84099697E-14  2.09164952E-09 -8.80165135E-06 

● テイラー展開 

ある関数 )(xf をべき級数で表すことができるとする． 

KK ++++++== ∑
∞

=

n
n

n
nn xcxcxcxccxxf 3

3
2

210
0

)(   （T-1） 

これは 0=x のまわりで展開している． )(xf は無限回微分可能とする．式（T-1）において

0=x とすると 

 )0(0 fc =  

式（T-1）を１回微分して 

KK +++++= −12
321

' 32)( n
n xncxcxccxf  

ここで， 0=x とすると 

 )0('
1 fc =  

同様に 

KK +−++⋅+= −2
32

'' )1(232)( n
n xcnnxccxf  

KK +−−++⋅= −3
3

''' )2)(1(23)( n
nxcnnncxf  

であるから 

)0(
2
1 ''

2 fc = ， )0(
23

1 '''
3 fc

⋅
= ，．．．， )0(

!
1 )(n

n f
n

c =  

したがって， 

KK +++++
′

+= n
n

x
n

f
x

f
x

f
x

f
fxf

!
)0(

!3
)0(

!2
)0(

!1
)0(

)0()(
)(

3
'''

2
''

 

となる．これをマクローリン級数という．べき級数の収束半径を rとして rx < で成立する．

)(xf を ax = のまわりで展開したとき 

となる．これをテイラー級数（展開）という． 

課題５ 区間 [ ]3,3− における 

 

0133)( 23 =+++= xxxxf  

 
の根を二分法で求めよ． 
 
 
 

（解法）関数 )(xf は区間 [ ]ba, で連続で，根が１つ存在するとき， )(af ， )(bf は異符合

になる． 0)()( <bfaf である．２点( ))(, afa ，( ))(, bfb を直線で結び，この直線とx軸と

の交点をcとすると 

)()(
)()(

afbf
abfbaf

c
−
−

=    (5-1) 

となる． 

0)()( <cfaf  ならば cb = ， )()( cfbf =  

0)()( <cfbf  ならば ca = ， )()( cfaf =  

の変換を繰り返して 

ε<−− ))(( cbac  

または   ε<)(cf  

となったとき，cを根とする．これを Regula falsi法という． 
 (5-1)式の代わりに 

2
ba

c
+

=  

とした方法を二分法という．収束条件として 

ε<− ab  

または   ε<)(cf  

を用いることが多い．この場合，反復計算回数nは 

ε<
−

n

ab

2
 

から計算することができる． 

（FORTRANプログラム例） 
 
c 
c      ex5  
c 
      f(x) = x**3 + 3.0*x**2 + 3.0*x + 1.0 
c 
      a  =  -3.0 
      b  =   3 .0  
c 
      fa = f(a) 
      do i=1,1000 
         c = (a + b)/2.0 
         fc = f(c)  
         write(6,*) i,c,fc 
         i f (  abs(b-a) .lt. 1.0e-5 ) then 
             s t o p  
         e n d i f  
         if( fa*fc .lt. 0.0 ) then 
             b  =  c  
         e l s e  
             a  =  c  
             f a  =  f c 
         e n d i f  
      enddo  
      stop 
      e n d  
 
 
（実行結果） 
 
1   0.   1.  
 2 -1.5 -0.125 
 3 -0.75  0.015625 
 4 -1.125 -0.001953125 
 5 -0.9375  0.000244140625 
 6 -1.03125 -3.05175781E-05 
 7 -0.984375  3.81469727E-06 
 8 -1.0078125 -4.76837158E-07 
 9 -0.99609375  5.96046448E-08 
 10 -1.00195312 -7.4505806E-09 
 11 -0.999023438  9.31322575E-10 
 12 -1.00048828 -1.16415322E-10 
 13 -0.999755859  1.45519152E-11 
 14 -1.00012207 -1.8189894E-12 
 15 -0.999938965  2.27373675E-13 



18 

 16 -1.00003052 -2.84217094E-14 
 17 -0.999984741  3.55271368E-15 
 18 -1.00000763 -4.4408921E-16 
 19 -0.999996185  5.55111512E-17 
 20 -1.00000191 -6.9388939E-18 
 21 -0.999999046  8.67361738E-19 
 
 
 
 
 
 
 

● 追加課題 

0)( 3 =−= xxxf  

の根をすべて求めるプログラムを作成せよ． 

課題６ 
 

∫=
π

0

sin xdxI  

 
を台形公式を用いて， xの幅 °20 ， °10 ， °5 ， °1 で計算値 I がどのようになるか示せ． 
 
 
 

（解法）区間 ( )nxxba ,),( 0≡ における定積分をn個の幅hの区間に分けて 

( ) ( )( )∑∫
−

=

++
≈=

1

0

1

2
)(

0

n

i

ii
x

x

hxfxf
dxxfI

n

 

で計算する（台形公式）．この式は 

( ) ( ) ( ) ( ) ( )( )nn xfxfxfxfxf
h

I +++++≈ −1210 2...22
2

 

となる． 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

( )1+kxf
( )kxf

1+kxkx0  

y

x

＊＊＊プログラム例＊＊＊ 
c 
c   ex6 
c 
      data pai / 3.1415927 / 
c 
      call integd(0.0,pai,pai*20.0/180.0,ans20) 
      call integd(0.0,pai,pai*10.0/180.0,ans10) 
      call integd(0.0,pai,pai*5.0/180.0,ans5) 
      call integd(0.0,pai,pai*1.0/180.0,ans1) 
      write(6,*) 'ans20 =',ans20 
      write(6,*) 'ans10 =',ans10 
      write(6,*) 'ans5  =',ans5 
      write(6,*) 'ans1  =',ans1 
      stop 
      end 
c*****integd 
      subroutine integd(x0,xn,h,ans) 
c 
      f(t) = sin(t) 
c      f(t) = cos(t) 
c      f(t) = 3.0/2.0*sin(t)*cos(t)**2.0 
c 
      ans = f(x0) 
      do x=x0+h,xn-h,h 
         ans = ans + 2.0*f(x) 
      enddo 
      ans = ans + f(xn) 
      ans = ans*h/2.0 
      return 
      end 
 
 
 
 
＊＊＊実行結果＊＊＊ 
 
ans20 =  1.86026323 
ans10 =  1.96461344 
ans5  =  1.99112535 
ans1  =  1.99995089 

課題７ 
 

∫=
1

0

dxxeI x  

 
をシンプソン（Simpson）の公式を用いて求めよ．分割数を 4=n ， 10=n ， 100=n とし，
積分結果を比較せよ．また，分割数を同じにしたとき，台形公式とシンプソン公式を用い
た場合の計算結果の違いを比較せよ． 
 
 

（解法）区間 ( )nxx ,0 における定積分を幅hの n2 個の区間に分ける． 

 
 
 
 
 
 
 
 
 
 
 
 
 
 

３点 ( ) ( ) ( )( )11 ,, +− kkk xfxfxf を通る２次曲線を ( ) ( ) cxxbxxay kk +−+−= 2
としたとき，

( )11 , +− kk xx 間の面積は 

( ) ( ) ( )

chah

xxcxxbxxaydxI
k

k

k

k

x

x
kkk

x

x
k

2
3
2

2
1

3
1

3

2
1

1

1

1

+=





 −+−+−==

+

−

+

−

∫
 

となる．ここで，  hxx kk −=−1 ， hxx kk +=+1  

とした．この関係を用いると 

( ) cbhahxf k +−=−
2

1 ， ( ) cxf k = ， ( ) cbhahxf k ++=+
2

1  

であるから， 

( ) ( ) ( )[ ]14
3 +++= kkkk xfxfxf
h

I  

y

( )1−kxf
 

1−kx

( )1+kxf
( )kxf

1+kxkx0

y

x

である．区間 ( )nxx ,0 における定積分 I は 

( ) ( ) ( ) ( )[ ] ( ) ( ) ( )[ ] ( )nnn xf
h

xfxfxf
h

xfxfxf
h

xf
h

I
3

...
3

4
...

3
2

3 1312420 +++++++++= −−

となる． 
 
 
 
＊＊＊プログラム例＊＊＊ 
c 
c   ex7 
c 
      call integs(0.0,1.0,4,ans1) 
      call integs(0.0,1.0,10,ans2) 
      call integs(0.0,1.0,100,ans3) 
      write(6,*) 'ans1 =',ans1 
      write(6,*) 'ans2 =',ans2 
      write(6,*) 'ans3 =',ans3 
      stop 
      end 
c*****integs 
      subroutine integs(x0,xn,n,ans) 
c 
      f(t) = t*exp(t) 
c 
      h = (xn-x0)/float(n) 
      ans = f(x0) 
      do i=1,n-1,2 
         x = x0 + h*float(i) 
         ans = ans + 4.0*f(x) 
      enddo 
      do i=2,n-2,2 
         x = x0 + h*float(i) 
         ans = ans + 2.0*f(x) 
      enddo 
      ans = ans + f(xn) 
      ans = ans*h/3.0 
      return 
      end 
 
 
＊＊＊実行結果＊＊＊ 
ans1 =  1.00016904 
ans2 =  1.00000441 
ans3 =  1.00000036 

課題８ 
 

関数 ( )21)( −−= xexf ， ππ ≤≤− x  

 

をフーリエ級数で表せ．フーリエ級数の次数が 3,2,1=M のときのフーリエ係数を求めよ． 

結果を図示せよ． 
 
 
 

（解法） ππ ≤≤− x で定義されている関数 )(xf が 

( )∑
∞

=

++=
1

0 sincos
2
1

)(
n

nn nxbnxaaxf  

の形に展開できたとき，右辺をフーリエ級数といい， 0a ， 1a ，．．． 1b ， 2b ，．．．をフーリエ

係数という．計算の便宜上 02
1

a とした．フーリエ係数は 

∫
−

=
π

ππ
nxdxxfan cos)(

1
  ,...)2,1,0( =n  

∫
−

=
π

ππ
nxdxxfbn sin)(

1
  ,...)3,2,1( =n  

となる． 
 
 
―――――――――――――― 参考 ―――――――――――――――――――――― 

mn ≠ で 0>+ mn のとき 
 

( ) ( ){ }

0)sin(
1

)sin(
1

2
1

coscos
2
1

coscos

=



 −

−
++

+
=

−++=

−

−−
∫∫

π

π

π

π

π

π

xmn
mn

xmn
mn

dxxmnxmnmxdxnx

 

 

( ) ( ){ }

0)sin(
1

)sin(
1

2
1

coscos
2
1

sinsin

=



 −

−
−+

+
−=

−−+−=

−

−−
∫∫

π

π

π

π

π

π

xmn
mn

xmn
mn

dxxmnxmnmxdxnx

1≥= mn のとき 
 

( ) π
π

π

π

π

π

π

=



 +=+=

−−−
∫∫ xnx

n
dxnxnxdxnx 2sin

2
1

2
1

12cos
2
1

coscos  

 

( ) π
π

π

π

π

π

π

=



 −−=−−=

−−−
∫∫ xnx

n
dxnxnxdxnx 2sin

2
1

2
1

12cos
2
1

sinsin  

 

 
 
＊＊＊プログラム例＊＊＊ 
c 
c   ex9 
c 
      real*4  an(20),bn(20) 
      real*4  fxx(4,100) 
      data pai /3.141592 / 
c 
      dx = 2.0*pai/100.0 
      do i=1,100 
         x = dx*float(i-1) - pai 
         fxx(1,i) = exp(-(x-1.0)**2.0) 
      enddo 
c 
      call fourier(-pai,pai,100,1,an,bn) 
      write(6,*) ' M = 1' 
         write(6,*) 'an(',0,') = ',an(1) 
      do i=2,2 
         write(6,*) 'an(',i-1,') = ',an(i),'   bn(',i-1,') = ',bn(i) 
      enddo 
      do i=1,100 
         x = dx*float(i-1) - pai 
         fxx(2,i) = an(1)/2.0 
         do n=2,2 
            fxx(2,i) = fxx(2,i) + an(n)*cos(float(n-1)*x) 
     -                          + bn(n)*sin(float(n-1)*x) 
         enddo 
      enddo 
c 
      call fourier(-pai,pai,100,2,an,bn) 
      write(6,*) ' M = 2' 
         write(6,*) 'an(',0,') = ',an(1) 
      do i=2,3 
         write(6,*) 'an(',i-1,') = ',an(i),'   bn(',i-1,') = ',bn(i) 

      enddo 
      do i=1,100 
         x = dx*float(i-1) - pai 
         fxx(3,i) = an(1)/2.0 
         do n=2,3 
            fxx(3,i) = fxx(3,i) + an(n)*cos(float(n-1)*x) 
     -                          + bn(n)*sin(float(n-1)*x) 
         enddo 
      enddo 
c 
      call fourier(-pai,pai,100,3,an,bn) 
      write(6,*) ' M = 3' 
         write(6,*) 'an(',0,') = ',an(1) 
      do i=2,4 
         write(6,*) 'an(',i-1,') = ',an(i),'   bn(',i-1,') = ',bn(i) 
      enddo 
      do i=1,100 
         x = dx*float(i-1) - pai 
         fxx(4,i) = an(1)/2.0 
         do n=2,4 
            fxx(4,i) = fxx(4,i) + an(n)*cos(float(n-1)*x) 
     -                          + bn(n)*sin(float(n-1)*x) 
         enddo 
      enddo 
c 
      do i=1,100 
         x = dx*float(i-1) - pai 
         write(7,*) x,fxx(1,i),fxx(2,i),fxx(3,i),fxx(4,i) 
      enddo 
c 
      stop 
      end 
c*****fourier 
      subroutine fourier(x0,xn,n,m,an,bn) 
      real*4  an(20),bn(20) 
c 
      h = (xn-x0)/float(n) 
c 
      do j=1,m+1 
         an(j) = fcos(j-1,x0) 
         do i=1,n-1,2 
            x = x0 + h*float(i) 
            an(j) = an(j) + 4.0*fcos(j-1,x) 
         enddo 
         do i=2,n-2,2 
            x = x0 + h*float(i) 

            an(j) = an(j) + 2.0*fcos(j-1,x) 
         enddo 
         an(j) = an(j) + fcos(j-1,xn) 
         an(j) = an(j)*h/3.0*2.0/(xn-x0) 
      enddo 
      do j=2,m+1 
         bn(j) = fsin(j-1,x0) 
         do i=1,n-1,2 
            x = x0 + h*float(i) 
            bn(j) = bn(j) + 4.0*fsin(j-1,x) 
         enddo 
         do i=2,n-2,2 
            x = x0 + h*float(i) 
            bn(j) = bn(j) + 2.0*fsin(j-1,x) 
         enddo 
         bn(j) = bn(j) + fsin(j-1,xn) 
         bn(j) = bn(j)*h/3.0*2.0/(xn-x0) 
      enddo 
c 
      return 
      end 
c*****fcos 
      function fcos(m,t) 
      fcos = exp(-(t-1.0)**2.0)*cos(float(m)*t) 
      return 
      end 
c*****fsin 
      function fsin(m,t) 
      fsin = exp(-(t-1.0)**2.0)*sin(float(m)*t) 
      return 
      end 
 
 
 
 
＊＊＊gnuplot 例＊＊＊ 
set xlabel 'X' 
set ylabel 'Y' 
plot 'fort.7' using 1:2 title 'exp(-x*x)' with lines,¥ 
     'fort.7' using 1:3 title 'M=1' with lines,¥ 
     'fort.7' using 1:4 title 'M=2' with lines,¥ 
     'fort.7' using 1:5 title 'M=3' with lines 
pause -1 

課題９ 

xy
dx
dy

−=  

の微分方程式をルンゲ・クッタ（Runge-Kutta）法で解け．ただし，計算区間は( )0.15,0.0
とし， 0.15)0( =y とする．ステップ幅 3.0=h ， 25.0=h ， 2.0=h について解け． 

 

 

（解法） 

   






=

=

00 )(

),(

yxy

yxf
dx
dy

     （9-1） 

を数値的に解くとき， nxx = において近似解 ny が求められていたとする．最も簡単な２段
公式ではステップ幅hだけ進んだ hxx nn +=+1 における値 1+ny を 

( )





++=
=

++=+

),(
),(

12

1

22111

kyhxhfk
yxhfk

kkyy

nn

nn

nn

βα

γγ

   （9-2） 

を用いて計算する．式（9-1）の真の解 )(xy のテイラー展開は 

{ }
)(

))(,())(,())(,(
2
1

))(,()(

)())(,(
2
1

))(,()()(

3

2

3'2

hO

xyxfxyxfxyxfhxyxhfxy

hOxyxfhxyxhfxyhxy

nnnnynnxnnn

nnnnnn

+

+++=

+++=+

 

となる． )( nn xyy = として式（9-2）に適用すると 

{ } )(),(),(),(

),()()(
3

2
2

211

hOyxfyxfyxfh

yxfhxyy

nnnnynnx

nnnn

+++

++=+

βαγ

γγ
 

となる． 2h の項まで一致するように 







==

=+

2
1

1

22

21

βγαγ

γγ
 

とおく． 

 2/12 =γ とした公式を改良オイラー公式またはホイン公式， 12 =γ のときを修正オイラ

ー公式という．１ステップ幅の中においてもとの方程式の右辺の値をいくつか計算し，そ

の重みつき平均で新しいステップにおける近似解を求める公式を総称してルンゲ・クッタ

公式，その解法をルンゲ・クッタ法という． 

●全微分と偏微分 

関数 ),( yxfz = において，点 ),( yxP を固定し xと yの増分 x∆ と y∆ が十分小さいとす

るとき， zの増分 z∆ は次のように表される． 

[ ] [ ]),(),(),(),(
),(),(

yxfyxxfyxxfyyxxf
yxfyyxxfz

−∆++∆+−∆+∆+=
−∆+∆+=∆

 

平均値の定理 

x
x

yxf
yxfyxxf ∆

∂
∂

=−∆+
),(

),(),(  

を用いると 

y
y

yxf
x

x
yxf

z ∆
∂

∂
+∆

∂
∂

=∆
),(),(
 

となる． x∆ と y∆ を無限小とするとき，微分dxとdyを用いて書くことができる． 

dy
y

yxf
dx

x
yxf

dz
∂

∂
+

∂
∂

=
),(),(
 

 

 

 

（プログラム例） 

c 

c     ex9 

c 

      y = 15.0 

      h = 0.3 

      do x=0.0,15.0,h 

         write(10,*) x,y 

         call runge(h,x,y) 

      enddo 

      y = 15.0 

      h = 0.25 

      do x=0.0,15.0,h 

         write(11,*) x,y 

         call runge(h,x,y) 

      enddo 

      y = 15.0 

      h = 0.2 

      do x=0.0,15.0,h 

         write(12,*) x,y 

         call runge(h,x,y) 

      enddo 

      stop 

      end 

c*****runge 

      subroutine runge(h,x,y) 

      real*4   k1, k2 

      k1 = h*(-x*y) 

      k2 = h*(-(x+h)*(y+k1)) 

      y = y + 0.5*(k1+k2) 

      return 

      end 

 

 

 

（gnuplot プログラム） 

set xrange [0:15] 

set yrange [0:15] 

set xlabel 'X' 

set ylabel 'Y' 

plot 'fort.10' title 'h=0.3' with lines,¥ 

     'fort.11' title 'h=0.25' with lines,¥ 

     'fort.12' title 'h=0.2' with lines 

pause –1 

 

 

 

―――――おまけプログラム――――――――――――――――――――――――――― 

c 

c     ex9-1 

c 

      write(6,*) 'h =' 

      read(5,*) h 

      y = 1.0 

      do x=0.0,15.0,h 

         write(10,*) x,y 

         y = y - h*sin(x) 

      enddo 

      y = 1.0 

      do x=0.0,15.0,h 

         write(11,*) x,y 

         call runge(h,x,y) 

      enddo 

      do x=0.0,15.0,0.1 

         write(12,*) x,cos(x) 

      enddo 

      stop 

      end 

c*****runge 

      subroutine runge(h,x,y) 

      real*4   k1, k2 

      k1 = -h*sin(x) 

      k2 = -h*sin(x+h) 

      y = y + 0.5*(k1+k2) 

      return 

      end 

 

 

 

（gnuplot プログラム） 

set xrange [0:6.28] 

set yrange [-1:1] 

set xlabel 'X' 

set ylabel 'Y' 

plot 'fort.10' title 'Euler' with lines,¥ 

     'fort.11' title 'Runge' with lines,¥ 

     'fort.12' title 'Exact' with lines 

pause -1 

―――――――――――――――――――――――――――――――――――――――― 

課題１０ 連成振動をルンゲ・クッタ（ Runge-Kutta）法で解け．また，ばねの定数 1c ， 2c ，

k  や質点の質量 1m ， 2m を変えて，計算してみよ． 

 
 
 
 
 
 
 
 
 
 
 
 
（解法）連立常微分方程式 

をルンゲ・クッタ（Runge-Kutta）法の２段公式で解くときは，課題９と同様に， 

とすればよい． 
 
 
 
 
 
 
 

k C1 C2 m2 m1 

 連成振動の運動方程式は 

である．速度を用いて 

の連立微分方程式を解くことになる． 
 
 
 
 
 
―――最もよく使われるルンゲ・クッタ（Runge-Kutta）法――― 

＊＊＊プログラム例＊＊＊ 
c 
c     ex10 
c 
      real*8   dt,pm,pn,pc,pk,x1,x2,v1,v2 
c 
      pm = 1.0 
      pn = 0.8 
      pc = 1.0 
      pk = 0.1 
      x1 = 0.1 
      x2 = 0.0 
      v1 = 0.0 
      v2 = 0.0 
      dt = 1.0d-3 
      tm = 0.0 
c 
      do i=1,100000 
         if( mod(i,100) .eq. 1 ) then 
             write(10,*) tm,x1,x2,v1,v2 
         endif 
         call runge(dt,pm,pn,pc,pk,x1,x2,v1,v2) 
         tm = tm + dt 
      enddo 
      stop 
      end 
c*****runge 
      subroutine runge(dt,pm,pn,pc,pk,x1,x2,v1,v2) 
      real*8   k11,k21,k31,k41 
      real*8   k12,k22,k32,k42 
      real*8   k13,k23,k33,k43 
      real*8   k14,k24,k34,k44 
      real*8   dt,pm,pn,pc,pk,x1,x2,v1,v2 
      k11 = dt/pn*(-(pc+pk)*x1+pk*x2) 
      k21 = dt*v1 
      k31 = dt/pm*(pk*x1-(pc+pk)*x2) 
      k41 = dt*v2 
      k12 = dt/pn*(-(pc+pk)*((x1+k11/2.0)+pk*(x2+k31/2.0))) 
      k22 = dt*(v1+k21/2.0) 
      k32 = dt/pm*(pk*(x1+k11/2.0)-(pc+pk)*(x2+k31/2.0)) 
      k42 = dt*(v2+k41/2.0) 
      k13 = dt/pn*(-(pc+pk)*((x1+k12/2.0)+pk*(x2+k32/2.0))) 
      k23 = dt*(v1+k22/2.0) 
      k33 = dt/pm*(pk*(x1+k12/2.0)-(pc+pk)*(x2+k32/2.0)) 
      k43 = dt*(v2+k42/2.0) 
      k14 = dt/pn*(-(pc+pk)*((x1+k13)+pk*(x2+k33))) 

      k24 = dt*(v1+k23) 
      k34 = dt/pm*(pk*(x1+k13)-(pc+pk)*(x2+k33)) 
      k44 = dt*(v2+k43) 
      x1 = x1 + (k21+2.0*k22+2.0*k23+k24)/6.0 
      x2 = x2 + (k41+2.0*k42+2.0*k43+k44)/6.0 
      v1 = v1 + (k11+2.0*k12+2.0*k13+k14)/6.0 
      v2 = v2 + (k31+2.0*k32+2.0*k33+k34)/6.0 
      return 
      end 
 
 
 
 
 
＊＊＊gnuplot のプログラム例＊＊＊ 
set xlabel 't' 
set ylabel 'X' 
plot 'fort.10' using 1:2 title 'x1' with lines,¥ 
     'fort.10' using 1:3 title 'x2' with lines 
pause -1 
set xlabel 't' 
set ylabel 'V' 
plot 'fort.10' using 1:4 title 'v1' with lines,¥ 
     'fort.10' using 1:5 title 'v2' with lines 
pause -1 

課題１１ r m離れたところに，大きさを無視できる電荷 q± がある．その周囲の電位を求

めよ． また，電気力線を示せ． 

 

 

（解法）ガウスの法則 

を用いて電位φ    φ−∇=E  

を求める．ここで，Eは電場， −−++ −= nqnqρ は電荷密度， 12
0 10854.8 −×=ε F/m は誘

を求めることにする．ポアソン方程式の一般形は， 

をラプラス方程式という． 

 電荷 tq を持った粒子は電気力線に沿って運動する．そこで，テスト粒子を系の中に入れ

粒子の軌跡を追うことにする．運動方程式は 

となる．また，電場Eを用いると 

である． φ−∇=E とルンゲ・クッタ（Runge-Kutta）法を用いて解いてみよ． 

 偏微分方程式を数値的に解く方法として差分法がある．関数 )(rf のテイラー展開は 

となる． 

x軸をn分割， y軸をm分割したときの，空間格子点をkとする．差分方程式は 

となる．境界条件は何らかの形で与えなくてはいけない． 

 差分方程式を反復法で解く場合（行列を作って解く方法もある）， 

を，すべての分割点に対して行う． ),( kk yxf が収束するまで繰り返し，収束したときの値

を解とする． 

 

 

 

＊＊＊Fortran プログラム例＊＊＊ 

c 

c     ex11 

c 

      real*4  fd(32,32) 

c 

      do i=1,32 

         do j=1,32 

            fd(i,j) = 0.0 

         enddo 

      enddo 

c 

      dx = 1.0 

      dy = 1.0 

      qc = 1.0 

      kx1 = 11 

      kx2 = 21 

      ky1 = 16 

      ky2 = 16 

c 

      tt = 0.0 

      do n=1,1000 

         tb = tt 

         tt = 0.0 

         do i=2,31 

         do j=2,31 

            fd(i,j) = fd(i-1,j)*dy*dy/2.0/(dx*dx+dy*dy) 

     -               +fd(i+1,j)*dy*dy/2.0/(dx*dx+dy*dy) 

     -               +fd(i,j-1)*dx*dx/2.0/(dx*dx+dy*dy) 

     -               +fd(i,j+1)*dx*dx/2.0/(dx*dx+dy*dy) 

            if( i .eq. kx1 .and. j .eq. ky1 ) then 

                fd(i,j) = fd(i,j) + qc*dx*dx*dy*dy/2.0/(dx*dx+dy+dy) 

            endif 

            if( i .eq. kx2 .and. j .eq. ky2 ) then 

                fd(i,j) = fd(i,j) - qc*dx*dx*dy*dy/2.0/(dx*dx+dy+dy) 

            endif 

            tt = tt + abs(fd(i,j)) 

         enddo 

         enddo 

         dtt = abs(tt-tb)/(abs(tt)+abs(tb)) 

         write(6,*) n,tt,dtt 

         if( dtt .lt. 1.0e-6 ) then 

             go to 900 

         endif 

      enddo 

  900 continue 

      do j=1,32 

         do i=1,32 

            write(10,*) fd(i,j) 

         enddo 

         write(10,*) 

      enddo 

      stop 

      end 

 

 

 

＊＊＊Gnuplot プログラム例＊＊＊ 

set xlabel 'X' 

set ylabel 'Y' 

set zlabel 'V' 

set contour 

splot 'fort.10' with lines 

pause -1 

課題１２ 温度 100℃で厚さ a m の板を 0℃の水中に入れたとき，板の内部の温度変化を
を求めよ． 

＊＊＊Fortran プログラム例＊＊＊ 
c 
c     ex12 
c 
      real*4  temp(64), temb(64) 
c 
      do i=2,63 
         temp(i) = 100.0 
      enddo 
         temp(1) = 0.0 
         temp(64) = 0.0 
c 
      dd = 0.5 
c 
      do n=1,2000 
         do i=2,63 
            temb(i) = temp(i) + 
     -                dd*(temp(i+1)-2.0*temp(i)+temp(i-1)) 
         enddo 
         do i=2,63 
            temp(i) = temb(i) 
         enddo 
         if( mod(n,100) .eq. 0 ) then 
            write(10,*) temp(1) 
            do i=2,63 
               write(10,*) temp(i) 
            enddo 
            write(10,*) temp(64) 
            write(10,*) 
            write(10,*) 
         endif 
      enddo 
  900 continue 
      stop 
      end 
 
 
 
＊＊＊Gnuplot プログラム例＊＊＊ 
set xlabel "x" 
set ylabel "T" 
set xrange [0:63] 
plot 'fort.10' with lines 
pause –1 

課題１３ １次元波動の伝播を数値的に解け．初期条件を変えて伝播の様子を調べてみよ． 
 
 
 
（解法）１次元の波動方程式 

の計算を繰り返し解く． 
 
 
 
 
 
＊＊＊Fortran プログラム例＊＊＊ 
c 
      real*4 u(3,128) 
      data dxdt2 / 0.1 / 
c 
      open(1,file='plot.out') 
         do i=1,128 
            read(1,*,end=100) j,u(1,i),u(2,i) 
         enddo 
         go to 200 
  100    continue 
         do i=1,128 
            u(1,i)=0.0 
            u(2,i)=0.0 
            if( i .ge. 8 .and. i .le. 20 ) then 
                u(1,i)=float(i-8)/12.0 
                u(2,i)=float(i-8)/12.0 
            endif 
            if( i .ge. 20 .and. i .le. 32 ) then 
                u(1,i)=1.0-float(i-20)/12.0 

                u(2,i)=1.0-float(i-20)/12.0 
            endif 
         enddo 
  200    continue 
      close(1) 
c 
      do i=2,127 
         u(3,i) = 2.0*u(2,i) - u(1,i) + 
     -             ( u(2,i+1) - 2.0*u(2,i) + u(2,i-1) )*dxdt2 
      enddo 
      do i=1,128 
         u(1,i) = u(2,i) 
         u(2,i) = u(3,i) 
      enddo 
c 
      open(1,file='plot.out') 
         do i=1,128 
            write(1,*) i,u(1,i),u(2,i) 
         enddo 
      close(1) 
c 
      stop 
      end 
 
 
 
＊＊＊Gnuplot プログラム例＊＊＊ 
# plot.gnu 
!rm plot.out 
!./a.out 
plot [1:128] [-1:1] "plot.out" using 1:3 with lines 
load "plot13a.gnu" 
 
 
 
＊＊＊Gnuplot プログラム例＊＊＊ 
# plot13a.gnu 
!./a.out 
replot 
reread 
 


